
 

  

 

Questo Plugin è nato con l’idea di comunicare in maniera 

rapida e sicura con il flexibowl tramite i robot FANUC. 

Il Plugin necessita della licenza "Fanuc User Socket 

Messaging" per il corretto funzionamento. 

CONFIGURAZIONE DELL’OPZIONE SOCKET 
MESSAGING  
Panoramica 
Al fine di usare Socket Messaging, è necessario configurare i parametri hardware and 

software del network: 

• Sul client, 

— L’indirizzo IP e nome del vostro server 

— La porta del Vostro server che volete usare per il socket messaging. 

Impostare un Client Tag 
È necessario configurare i tag del cliente se volete usare per il socket messaging.  

Nota Se i tag cliente che volete usare sono usati da un protocollo diverso dal TCP/IP, 

dovrete indeterminate i tag prima che possano essere usati per il socket messaging. 
 

Procedure per installare un ClientTag 
Condizioni 

• Il tag che volete installare non deve essere configurato per essere usato da un altro 

dispositivo sulla vostra rete. 

Passaggi 
1. Cold start del controller. 

a. Sulla teach pendant, premere a lungo I tasti SHIFT e RESET. Oppure, sul pannello 

operatore, premere a lungo RESET. 

b. Mentre state premendo SHIFT e RESET sulla teach pendant (o RESET sul pannello 

operatore), accendere l’alimentazione. 

c. Rilasciare tutti i tasti. 

2. Sulla teach pendant, premere MENUS. 

3. Selezionare SETUP. 

4. Premere F1, [TYPE]. 

5. Selezionare Host Comm. 

6. Premere F4, [SHOW]. 

7. Scegliere Clients. 

8. Muovere il cursore sul tag che si vuole impostare per il Socket Messaging, e premere 

F3, DETAIL. 

Vedrete una schermata similie alla seguente. 

 



 

  

 

9. Muovere il cursore su Protocol Name item, e premere F4, [CHOICE]. 

10. Selezionare SM. 

11. Muovere il cursore su  Startup State item, premere F4, [CHOICE], and scegliere 

DEFINE. 

12. Muovere il cursore su Server IP/Hostname item, e premere ENTER. 

13. Scrivere l’IP del remote host server in hostname o indirizzo IP se volete usarlo per il 

socket messaging. 

Nota Se non state usando DNS, dovete aggiungere il remote host e il suo indirizzo IP nel 

host entry table. 

14. Premere F2, [ACTION], e selezionare DEFINE. 

15. Impostare le variabili di sistema:  

a. Premere MENUS. 

b. Selezionare NEXT. 

c. Selezionare SYSTEM, e premere F1, [TYPE]. 

d. Selezionare Variables. 

e. Muovere il cursore su $HOSTC_CFG, e premere ENTER. 

f. Muovere il cursore sulla struttura corrispondente al tag selezionato al Passaggio 8 . Per 

esempio, se sis ta impostando il tag C3, muovere il cursore su structure element [3], come 

da foto seguente. 

 
g. Premere ENTER. Vedrete una schermata similie alla seguente. 

 
h. Muovere il cursore su $SERVER_PORT. Scrivere il nome della porta server TCP/IP che 

volete usare per il socket messaging. Il client tag è ora pronto per essere usato con un 

programma KAREL. 

A seguire si indica come impostare il Cliente e il codice Karel per comunicare con il 

FlexiBowl. Tale script puo essere richiamato per il movimento the FlexiBowl or oppure messo 

in un task parallelo, e regolare l’esecuzione tramite un semaforo. Il codice riceverà un 

comando da eseguire, e restituirà una stringa con la risposta dal FlexiBowl 

STEP 2: Configurazione del Host Client C7 per comunicare con l’indirizzo del FlexiBowl: 

192.168.1.10 

Configurazione della variabile di sistema 

STEP 4: File Karel 

Una volta che il client è configurato, si deve inserire il file FLB_PLUGIN.PC nel controller. 

Per fare ciò, copiare il file su una chiavetta USB e inserirla nella porta USB della pendant. 

Gestire i comandi del programma Tp 

Una volta che il client è configurato, dopo aver importato il file Flb_Plugin.pc, sarà possibile 

muovere il FlexiBowl o fare diagnosi dal programma Tp, la risposta del FlexiBowl sarà inserita 

nella String [10], editabile via Karel. 

Esmpio di codice KAREL: 

 

PROGRAM Flb_Plugin 
 
VAR 
PingAddress:BOOLEAN 
return_string : STRING[128] 
status : INTEGER 
moving:INTEGER 
cmd_type: INTEGER 
cmd_int_val:INTEGER 
cmd_real_val:REAL 
cmd_str_val:STRING[128] 
 
command_Str:STRING[128] 
file_var : FILE 
tmp_str : STRING[128] 
tmp_str1:STRING[128] 
tmp_str2:STRING[128] 
entry : INTEGER 
 
 
ROUTINE Ping 
BEGIN 
 
 PingAddress=FALSE 
 MSG_DISCO('C7:',status) 
 MSG_CONNECT('C7:',status) 
 IF(status=0) THEN 
  PingAddress=TRUE 
 else 
  PingAddress=FALSE 
 ENDIF 
 MSG_DISCO('C7:',status) 
   
END Ping 
 
----/////////////////////////////////////////////////////////////////////////////////////////////////////////////////////// 
---////////////////////////////////////////////////SENDCOMMAND//////////////////////////////////////////// 
--/////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
  
ROUTINE SendCommand(CommandExecute:STRING):STRING 
 
 
BEGIN 
return_string='FAIL'; 
command_Str=CommandExecute 

CLR_IO_STAT(file_var) 
status = IO_STATUS(file_var) 
IF status = 0 THEN  
 --Write 
 tmp_str=CHR(0)+CHR(7)+ command_Str + CHR(13) 
 WRITE file_var (tmp_str) 
 IF( IO_STATUS(file_var)<>0) THEN 
  return_string='FAIL' 
  WRITE ('Status var OUT fail',CR) 
  go to end_it  
 ENDIF 
  
 --READ 
 entry=0 
 WHILE entry <1 DO 
  BYTES_AHEAD(file_var, entry, status) 
 ENDWHILE 
  
 STATUS = IO_STATUS(file_var) 
 IF(STATUS<>0)THEN 
  return_string='FAIL' 
  WRITE ('Status var IN fail',CR) 
  go to end_it  
 ENDIF 
 IF (entry>0) THEN 
  READ file_var (tmp_str1::entry) 
  tmp_str2=SUB_STR(tmp_str1,2,entry-1) 
  return_string=tmp_str2 
 ELSE 
  return_string='FAIL' 
  go to end_it 
 ENDIF 
ELSE 
 WRITE ('Status var fail',CR) 
 return_string='FAIL' 
 go to end_it  
ENDIF 
  
end_it:: 
RETURN(return_string) 
END SendCommand 
----/////////////////////////////////////////////////////////////////////////////////////////////////////////////////////// 
 



 

  

 

BEGIN  
 GET_TPE_PRM(1,cmd_type,cmd_int_val,cmd_real_val,cmd_str_val,S
TATUS) 
 IF(STR_LEN(cmd_str_val)>0) THEN 
  command_Str=cmd_str_val 
 else 
  return_string='FAIL' 
  go to end_it2 
   
 ENDIF 
 --setto i parametri di connessione 
 SET_FILE_ATR(file_var, ATR_IA) 
 MSG_DISCO('C7:', status) 
 MSG_CONNECT('C7:',status) 
 IF(status <> 0) THEN 
  return_string='FAIL' 
  go to end_it2 
 ENDIF 
 OPEN FILE file_var ('RW','C7:') 
 return_string=SendCommand(command_Str) 
 IF((INDEX(return_string,'%')<>0)  

AND (INDEX(command_Str,'Q')<>0)) THEN 
  -----wait move 
  moving=1 
  WHILE (moving = 1) DO 
                 command_Str='RS' 
   return_string=SendCommand(command_Str) 
   IF(INDEX(return_string,'F')>0) THEN 
    moving = 1 
   else 
    moving = 0 
   ENDIF 
   DELAY(50) 
   ENDWHILE 
   return_string='Done' 
 else 
    return_string=return_string 
 ENDIF 
end_it2::  
 CLOSE FILE file_var 
 MSG_DISCO('C7:',status) 
 
 --write the return on string 10 
 SET_STR_REG(10,return_string,status) 
END Flb_Plugin 



 

  

 

 
 

   
 

 

Comandi      Descrizione 

QX2 Move 

QX3 Move-Flip 

QX4 Move-Flip-Blow 

QX5 Move-Blow 

QX6 Shake 

QX7 Light on 

QX8 Light off 

QX9 Flip 

QX10 Blow 

QX11 Quick Emptying Option 

QX12 Reset Alarm 

 

Lista dei comandi:  



 


